



2022 로보인 학기중 프로젝트

2자유도 로봇팔

고경환, 정래현



목차

A table of Contents

#0, 프로젝트 목표

#1, 역기구학

#2, 회로도 및 하드웨어

#3, 코드

#4, 작동영상

#5, 어려웠던 점




프로젝트 목표

- ① shoulder와 elbow의 각을 입력해 움직이게 하고 베이스로부터 수평거리와 수직거리를 출력
- ② 베이스로부터 수평거리와 수직거리를 입력해 로봇팔의 end-effector가 해당 위치에 도달하도록 함
- + 추가 달성 목표 1 : base의 rotate 방향 모터 제어
- + 추가 달성 목표 2 : 항상 집게 부분 손목 각도가 바닥과 평행하게 유지되도록 함
- + 추가 달성 목표 3 : 집게를 제어해 물체를 집을 수 있도록 함

end-effector까지의 거리를 알고,
두 팔(forearm, arm)의 길이를
알기 때문에 세 변의 길이를 앎

=>

삼각형으로 제 2 코사인 법칙을
이용하여 각도를 구함



$\theta_3 = \text{elbow angle}$


$$\cos \theta_3 = \frac{f_a^2 + a^2 - D^2}{2 \cdot f_a \cdot a}$$



$$\cos \theta_2 = \frac{D^2 + a^2 - f_a^2}{2 \cdot D \cdot a}$$

$$\cos(\theta_1 + \theta_2) = \frac{H}{D}$$

$$\cos \theta_4 = \frac{D^2 + f_a^2 - a^2}{2 \cdot D \cdot a}$$

$$\cos \theta_5 = \frac{V}{D}$$





```
#include <Servo.h>

Servo shoulder;
Servo elbow;
Servo wrist;

int shoulderPin = 5;
int elbowPin = 6;
int wristPin = 7;
int shoulderAngle = 90;
int elbowAngle = 90;
int wristAngle = 90;

void setup() {
  Serial.begin(9600);

  shoulder.attach(shoulderPin);
  elbow.attach(elbowPin);
  wrist.attach(wristPin);

  shoulder.write(shoulderAngle);
  elbow.write(elbowAngle);
  wrist.write(wristAngle);
}
```

```
void loop() {
  if(Serial.available()){
    shoulderAngle = Serial.parseInt();
    elbowAngle = Serial.parseInt();
    wristAngle = Serial.parseInt();
  }
  shoulder.write(shoulderAngle);
  elbow.write(elbowAngle);
  wrist.write(wristAngle);
  Serial.print("Shoulder각도 : ");
  Serial.print(shoulderAngle);
  Serial.print("  Elbow각도 : ");
  Serial.print(elbowAngle);
  Serial.print("  wrist각도 : ");
  Serial.println(wristAngle);
}
```

```
#include <Servo.h>
#define _USE_MATH_DEFINES
#include <math.h>

Servo shoulder;
Servo elbow;
Servo rotate;
Servo hand;
Servo wrist;

int rotatePin = 4;
int shoulderPin = 5;
int elbowPin = 6;

int wristPin = 7;
int handPin = 8;

double rotateAngle = 90;
double shoulderAngle = 90;
double elbowAngle = 90;
double wristAngle = 180;

int handOpen = 170;
int handClose = 80;
int isOpen = 1;

double armLength = 12;
double foreArmLength = 12;

double horizontal = 12;
double vertical = 12;
double distance = 0;
```

```
void setup() {
    Serial.begin(9600);
    shoulder.attach(shoulderPin);
    elbow.attach(elbowPin);
    rotate.attach(rotatePin);
    hand.attach(handPin);
    wrist.attach(wristPin);

    shoulder.write(shoulderAngle);
    elbow.write(elbowAngle);
    rotate.write(rotateAngle);
    hand.write(handOpen);
    wrist.write(wristAngle);
}
```



```
void loop() {
    if(Serial.available()){
        rotateAngle = Serial.parseInt();
        horizontal = Serial.parseInt();
        vertical = Serial.parseInt();
        isOpen = Serial.parseInt();
    }

    distance = sqrt( pow(horizontal,2) + pow(vertical,2) );

    elbowAngle = 180 / M_PI * acos((pow(armLength, 2) + pow(foreArmLength, 2) - pow(distance, 2)) / (2 * armLength * foreArmLength));

    shoulderAngle = 180 / M_PI * ( acos(horizontal / distance) - acos( (pow(armLength, 2) + pow(distance, 2) - pow(foreArmLength, 2)) / (2 * armLength * distance)) );

    wristAngle = 180 / M_PI * ( ( acos(vertical / distance) - acos( (pow(foreArmLength, 2) + pow(distance, 2) - pow(armLength, 2)) / (2 * foreArmLength * distance)) ) + M_PI/2);

    if (isOpen == 1) {
        hand.write(handOpen);
    }
    if (isOpen == 0) {
        hand.write(handClose);
    }

    shoulder.write(shoulderAngle);
    elbow.write(elbowAngle);
    rotate.write(rotateAngle);
    wrist.write(wristAngle);

    Serial.print("Rotate각도 : ");
    Serial.print(rotateAngle);
    Serial.print(" Shoulder각도 : ");
    Serial.print(shoulderAngle);
    Serial.print(" Elbow각도 : ");
    Serial.print(elbowAngle);
    Serial.print(" Wrist각도 : ");
    Serial.print(wristAngle);
    Serial.print(" hand on ? : ");
    Serial.println(isOpen);
}
```

<https://cafe.naver.com/yonseiunivroboin/247>

어려웠던 점

1. 3D프린터 서포트가 떨어지지 않아 부품 재 출력

2. hand 부분 서보모터와 연결하기 어려운 곳 3D 프린터 출력물을 드레멜로 갈아서 끼움